Knot Invariants from Finite Dimensional Integration

Thanks for having me in Groningen!

ωεβ:=http://drorbn.net/g24

Abstract. For the purpose of today, an "I-Type Knot Invariant" is a knot invariant computed from a knot diagram by integrating the exponential of a *perturbed Gaussian* Lagrangian which is a sum over the features of that diagram (crossings, edges, faces) of locally defined quantities, over a product of finite dimensional spaces associated to those same features.

joint with R. van der Veen

Q. Are there any such things? **A.** Yes.

Q. Are they any good? **A.** They are the strongest we know per CPU cycle, and are excellent in other ways too.

Q. Didn't Witten do that back in 1988 with path integrals?

A. No. His constructions are infinite dimensional and far from rigorous.

Q. But integrals belong in analysis!

A. Ours only use squeaky-clean algebra.

Continues Rozansky  Overbay 

The $sl_2^{/\epsilon^2}$ Example. With T an indeterminate and with $\epsilon^2 = 0$:



$$\longrightarrow Z = \oint_{\mathbb{R}^{14}_{p_ix_i}} \mathcal{L}(X^+_{15})\mathcal{L}(X^+_{62})\mathcal{L}(X^+_{37})\mathcal{L}(C_4^{-1})$$

measure on $\mathbb{R}$ is $(2\pi)^{-1/2}\cdot$*standard*

where $\mathcal{L}(X^s_{ij}) = T^{s/2}\mathbb{e}^{L(X^s_{ij})}$ and $\mathcal{L}(C^\varphi_i) = T^{\varphi/2}\mathbb{e}^{L(C^\varphi_i)}$, and

$$L(X^s_{ij}) = x_i(p_{i+1}-p_i) + x_j(p_{j+1}-p_j) + (T^s-1)x_i(p_{i+1}-p_{j+1}) + \frac{\epsilon s}{2}\left(x_i(p_i-p_j)\begin{pmatrix}(T^s-1)x_ip_j\\ +2(1-x_jp_j)\end{pmatrix}-1\right)$$

$$L(C^\varphi_i) = x_i(p_{i+1}-p_i) + \epsilon\varphi(1/2 - x_ip_i)$$

So $Z = T\oint \mathbb{e}^{L(\circledS)}dp_1\dots dp_7dx_1\dots dx_7$, where $L(\circledS) =$

$$\sum_{i=1}^7 x_i(p_{i+1}-p_i)+(T-1)(x_1(p_2-p_6)+x_6(p_7-p_3)+x_3(p_4-p_8)) + \frac{\epsilon}{2}\begin{pmatrix} x_1(p_1-p_5)\,((T-1)x_1p_5+2(1-x_5p_5))-1\\ +x_6(p_6-p_2)\,((T-1)x_6p_2+2(1-x_2p_2))-1\\ +x_3(p_3-p_7)\,((T-1)x_3p_7+2(1-x_7p_7))-1\\ +2x_4p_4-1\end{pmatrix},$$

and so $Z = (T-1+T^{-1})^{-1}\exp\left(\epsilon\cdot\frac{(T-2+T^{-1})(T+T^{-1})}{(T-1+T^{-1})^2}\right) = \Delta^{-1}\exp\left(\epsilon\cdot\frac{(T-2+T^{-1})\rho_1}{\Delta^2}\right)$. Here Δ is Alexander's polynomial and ρ_1 is Rozansky-Overbay's polynomial [R1]–[R3], [Ov, BV1, BV2].

Theorem. Z is a knot invariant.

Guido Fubini 

Proof. Use Fubini (details later).



To Do. • Human-hard but computer-very-easy (poly time!).
• Strong! • Details of the proof. • Where is it coming from?
• A philosophical point: "Universal invariants" are qualitatively better than representation theory ones.

Knots.



invariants ⟶ Something simple: numbers, polynomials, matrices, etc.

The Good. 1. At the centre of low dimensional topology.
2. "Invariants" connect to pretty much all of algebra.

The Agony. 1&2 don't talk to each other.

- Not enough topological applications for all these invariants.
- The fancy algebra doesn't arise naturally within topology.

⟹ We're still missing something about the relationship between knots and algebra.

(Alternative) Gaussian Integration.

Gauss 

Goal. Compute (if convergent) $\displaystyle\int_{\mathbb{R}^n} dx\exp\left(-\frac12 a^{ij}x_ix_j + V(x)\right)$.

Solution. Set $\displaystyle\mathcal{Z}_\lambda(x) := \lambda^{n/2}\int_{\mathbb{R}^n} dy\exp\left(-\frac{1}{2\lambda}a^{ij}y_iy_j + V(x+y)\right)$.

Then $\mathcal{Z}_1(0)$ is what we want, $\mathcal{Z}_0(x) = (\det A)^{-1/2}\exp V(x)$, and with g_{ij} the inverse matrix of a^{ij} and noting that under the dy integral $\partial_y = 0$,

$$\frac12 g_{ij}\partial_{x_i}\partial_{x_j}\mathcal{Z}_\lambda(x)$$
$$= \frac12\int_{\mathbb{R}^n} dy\, g_{ij}(\partial_{x_i}-\partial_{y_i})(\partial_{x_j}-\partial_{y_j})\exp\left(-\frac{1}{2\lambda}a^{ij}y_iy_j + V(x+y)\right)$$
$$= \frac{1}{2\lambda^2}\int_{\mathbb{R}^n} dy\left(g_{ij}a^{ii'}a^{jj'}y_{i'}y_{j'} + \lambda g_{ij}a^{ji}\right)\exp\left(-\frac{1}{2\lambda}a^{ij}y_iy_j + V(x+y)\right)$$
$$= \frac{1}{2\lambda^2}\int_{\mathbb{R}^n} dy\left(a^{ij}y_iy_j + \lambda n\right)\exp\left(-\frac{1}{2\lambda}a^{ij}y_iy_j + V(x+y)\right)$$
$$= \partial_\lambda\mathcal{Z}_\lambda(x).$$

Hence $$(*)\qquad \partial_\lambda\mathcal{Z}_\lambda(x) = \frac12 g_{ij}\partial_{x_i}\partial_{x_j}\mathcal{Z}_\lambda(x),$$

and therefore $$\mathcal{Z}_\lambda(x) = (\det A)^{-1/2}\exp\left(\frac{\lambda}{2}g_{ij}\partial_{x_i}\partial_{x_j}\right)\exp V(x).$$

We've just witnessed the birth of "Feynman Diagrams".



Feynman

Even better. With $Z_\lambda := \log(\sqrt{\det A}\,\mathcal{Z}_\lambda)$, by a simple substitution into (*), we get the "Synthesis Equation":

$$Z_0 = V, \qquad \partial_\lambda Z_\lambda = \frac12\sum_{i,j=1}^n g_{ij}\left(\partial_{x_i,x_j}Z_\lambda + (\partial_{x_i}Z_\lambda)(\partial_{x_j}Z_\lambda)\right),$$

an ODE (in λ) whose solution is pure algebra.

Picard Iteration (used to prove the existence and uniqueness of solutions of ODEs). To solve $\partial_\lambda f_\lambda = F(f_\lambda)$ with a given f_0, start with f_0, iterate $f\mapsto f_0+\int_0^\lambda F(f_\lambda)d\lambda$, and seek a fixed point. In our cases, it is always reached after finitely many iterations!

Picard

Definition. $\oint$: The result of this process, ignoring the convergence of the actual integral.

Strong. A faster program to compute ρ_1 is available at [BV2]. With it we find that the pair (Δ,ρ_1) attains 53,684 distinct values on the 59,937 prime knots with up to 14 crossings (a deficit of 6,253), whereas the pair (HOMFLYPT polynomial, Khovanov Homology) attains only 49,149 distinct values on the same knots (a deficit of 10,788).

In as much as we know the pair (Δ,ρ_1) is the strongest knot invariant that can be computed in polynomial time (and hence, even for very large knots).

Preliminaries

This is IType.nb of ωεβ/ap.

```
☺ Once[<< KnotTheory`; << Rot.m];
```

```
Loading KnotTheory` version
  of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.

Loading Rot.m from
  http://drorbn.net/AP/Talks/Groningen-240530
  to compute rotation numbers.
```

```
☺ CF[ω_. ℰ_𝔼] := CF[ω] × CF /@ ℰ;
CF[ℰ_List] := CF /@ ℰ;
CF[ℰ_] := Module[{vs, ps, c},
   vs = Cases[ℰ, (x | p | ξ | π)__, ∞] ⋃ {x, p, ϵ};
   Total[CoefficientRules[Expand[ℰ], vs] /.
      (ps_ → c_) ↦ Factor[c] (Times @@ vs^ps)] ];
```

Integration

Using Picard Iteration!

```
☺ 𝔼 /: 𝔼[A_] × 𝔼[B_] := 𝔼[A + B];

☺ $π = Identity; (* hacks in pink *)

☺ Unprotect[Integrate]; (* keys in yellow *)
∫ω_. 𝔼[L_] ⅆ(vs_List) :=
  Module[{n, L0, Q, Δ, G, Z0, Z, λ, DZ, FZ, a, b},
   n = Length@vs; L0 = L /. ϵ → 0;
   Q = Table[(-∂_{vs[[a]],vs[[b]]} L0) /. Thread[vs → 0] /.
       (p | x)__ → 0, {a, n}, {b, n}];
   If[(Δ = Det[Q]) == 0, Return@"Degenerate Q!"];
   Z = Z0 = CF@$π[L + vs.Q.vs / 2]; G = Inverse[Q];
   DZ_a_ := ∂_{vs[[a]]} Z; DZ_{a_,b_} := ∂_{vs[[b]]} DZ_a;
   FZ := CF@$π[1/2 Σ_{a=1}^{n} Σ_{b=1}^{n} G[[a, b]] (DZ_{a,b} + DZ_a DZ_b)];
   FixedPoint[(Z = Z0 + ∫_0^λ FZ ⅆλ) &, Z];
   PowerExpand@Factor[ω Δ^(-1/2)] ×
     𝔼[CF[Z /. λ → 1 /. Thread[vs → 0]]]];
Protect[Integrate];
```

```
☺ ∫𝔼[- μ x^2 / 2 + ⅈ ξ x] ⅆ{x}
```

$$\frac{\mathbb{E}\left[-\frac{\xi^2}{2\mu}\right]}{\sqrt{\mu}}$$

```
☺ L = -1/2 {x1, x2}.(a b; b c).{x1, x2} + {ξ1, ξ2}.{x1, x2};
Z12 = ∫𝔼[L] ⅆ{x1, x2}
```

$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2}{2\left(-b^2+a\,c\right)}+\frac{b\,\xi_1\,\xi_2}{b^2-a\,c}+\frac{a\,\xi_2^2}{2\left(-b^2+a\,c\right)}\right]}{\sqrt{-b^2+a\,c}}$$

```
☺ {Z1 = ∫𝔼[L] ⅆ{x1}, Z12 == ∫Z1 ⅆ{x2}}
```

$$\left\{\frac{\mathbb{E}\left[-\frac{\left(-b^2+a\,c\right)x_2^2}{2\,a}-\frac{b\,x_2\,\xi_1}{a}+\frac{\xi_1^2}{2\,a}+x_2\,\xi_2\right]}{\sqrt{a}},\ \text{True}\right\}$$

```
☺ $π = Normal[# + O[ϵ]^13] &; ∫𝔼[-ϕ^2 / 2 + ϵ ϕ^3 / 6] ⅆ{ϕ}
```

$$\mathbb{E}\left[\frac{5\,\epsilon^2}{24}+\frac{5\,\epsilon^4}{16}+\frac{1105\,\epsilon^6}{1152}+\frac{565\,\epsilon^8}{128}+\frac{82825\,\epsilon^{10}}{3072}+\frac{19675\,\epsilon^{12}}{96}\right]$$

From https://oeis.org/A226260:


THE ON-LINE ENCYCLOPEDIA OF INTEGER SEQUENCES ®

founded in 1964 by N. J. A. Sloane

Search Hints

(Greetings from The On-Line Encyclopedia of Integer Sequences!)

A226260 Numerators of mass formula for connected vacuum graphs on 2n nodes for a phi^3 field theory.

1, 5, 5, 1105, 565, 82825, 19675, 1282031525, 80727925, 1683480621875, 13209845125, 2239646759308375, 19739117098375, 6320791709083309375, 32468078556378125, 38362676768845045751875, 2813657784050329731 25, 282465074708942558615248437 5, 7766321570341167127343 75 (list; graph; refs; listen; history; text; internal format)

The Right-Handed Trefoil

```
☺ K = Mirror@Knot[3, 1]; Features[K]
```

Features[7, $C_4[-1]\ X_{1,5}[1]\ X_{3,7}[1]\ X_{6,2}[1]$]

```
☺ ℒ[X_{i_,j_}[s_]] := T^(s/2) 𝔼[
     x_i (p_{i+1} - p_i) + x_j (p_{j+1} - p_j) +
      (T^s - 1) x_i (p_{i+1} - p_{j+1}) +
      (ϵ s / 2) ×
       (x_i (p_i - p_j) ((T^s - 1) x_i p_j + 2 (1 - x_j p_j)) - 1)];
ℒ[C_i_[φ_]] := T^(φ/2) 𝔼[x_i (p_{i+1} - p_i) + ϵ φ (1/2 - x_i p_i)]
ℒ[K_] := CF[ℒ /@ Features[K][[2]]]
vs[K_] :=
 Join @@ Table[{p_i, x_i}, {i, Features[K][[1]]}]
```

```
☺ {vs[K], ℒ[K]}
```

$$\{\{p_1, x_1, p_2, x_2, p_3, x_3, p_4, x_4, p_5, x_5, p_6, x_6, p_7, x_7\},$$
$$T\,\mathbb{E}\Big[-2\,\epsilon - p_1 x_1 + \epsilon\, p_1 x_1 + T\, p_2 x_1 - \epsilon\, p_5 x_1 + (1-T)\, p_6 x_1 + \frac{1}{2}(-1+T)\,\epsilon\, p_1 p_5 x_1^2 + \frac{1}{2}(1-T)\,\epsilon\, p_5^2 x_1^2 - p_2 x_2 + p_3 x_2 - p_3 x_3 + \epsilon\, p_3 x_3 + T\, p_4 x_3 - \epsilon\, p_7 x_3 + (1-T)\, p_8 x_3 + \frac{1}{2}(-1+T)\,\epsilon\, p_3 p_7 x_3^2 + \frac{1}{2}(1-T)\,\epsilon\, p_7^2 x_3^2 - p_4 x_4 + \epsilon\, p_4 x_4 + p_5 x_4 - p_5 x_5 + p_6 x_5 - \epsilon\, p_1 p_5 x_1 x_5 + \epsilon\, p_5^2 x_1 x_5 - \epsilon\, p_2 x_6 + (1-T)\, p_3 x_6 - p_6 x_6 + \epsilon\, p_6 x_6 + T\, p_7 x_6 + \epsilon\, p_2^2 x_2 x_6 - \epsilon\, p_2 p_6 x_2 x_6 + \frac{1}{2}(1-T)\,\epsilon\, p_2^2 x_6^2 + \frac{1}{2}(-1+T)\,\epsilon\, p_2 p_6 x_6^2 - p_7 x_7 + p_8 x_7 - \epsilon\, p_3 p_7 x_3 x_7 + \epsilon\, p_7^2 x_3 x_7\Big]\}$$

```
☺ $π = Normal[# + O[ϵ]^2] &; ∫ℒ[K] ⅆ(vs@K)
```

$$-\frac{ⅈ\,T\,\mathbb{E}\left[-\frac{(-1+T)^2\left(1+T^2\right)\epsilon}{\left(1-T+T^2\right)^2}\right]}{1-T+T^2}$$

Invariance Under Reidemeister 3



☺
```
lhs = ∫(ℒ /@ (X_{i,j}[1] X_{i+1,k}[1] X_{j+1,k+1}[1]))
    d{p_{i+1}, p_{j+1}, p_{k+1}, x_{i+1}, x_{j+1}, x_{k+1}};
rhs = ∫(ℒ /@ (X_{j,k}[1] X_{i,k+1}[1] X_{i+1,j+1}[1]))
    d{x_{i+1}, p_{i+1}, p_{j+1}, p_{k+1}, x_{j+1}, x_{k+1}};
lhs === rhs
```
💻
```
False
```

Invariance Under Reidemeister 3, Take 2

☺
```
lhs = ∫(ℒ /@ (X_{i,j}[1] X_{i+1,k}[1] X_{j+1,k+1}[1]))
    d{x_i, x_j, x_k, p_{i+1}, p_{j+1}, p_{k+1}, x_{i+1}, x_{j+1}, x_{k+1}};
rhs = ∫(ℒ /@ (X_{j,k}[1] X_{i,k+1}[1] X_{i+1,j+1}[1]))
    d{x_i, x_j, x_k, x_{i+1}, p_{i+1}, p_{j+1}, p_{k+1}, x_{j+1}, x_{k+1}};
lhs === rhs
```
💻
```
True
```
☺
```
lhs
```
💻
```
Degenerate Q!
```

Invariance Under Reidemeister 3, Take 3

☺
```
lhs =
  ∫(𝔼[𝕚 π_i p_i + 𝕚 π_j p_j + 𝕚 π_k p_k] ×
     ℒ /@ (X_{i,j}[1] X_{i+1,k}[1] X_{j+1,k+1}[1]))
   d{p_i, p_j, p_k, x_i, x_j, x_k, p_{i+1}, p_{j+1}, p_{k+1}, x_{i+1},
     x_{j+1}, x_{k+1}};
rhs =
  ∫(𝔼[𝕚 π_i p_i + 𝕚 π_j p_j + 𝕚 π_k p_k] ×
     ℒ /@ (X_{j,k}[1] X_{i,k+1}[1] X_{i+1,j+1}[1]))
   d{p_i, p_j, p_k, x_i, x_j, x_k, p_{i+1}, p_{j+1}, p_{k+1}, x_{i+1},
     x_{j+1}, x_{k+1}};
lhs == rhs
```
💻
```
True
```
☺
```
lhs
```
💻

$$T^{3/2}\,\mathbb{E}\Big[-\frac{3\epsilon}{2} + i\,T^2 p_{2+i}\pi_i - i\,(-1+T)\,T\,p_{2+j}\pi_i + i\,T^2\epsilon\, p_{2+j}\pi_i - i\,(-1+T)\,p_{2+k}\pi_i + i\,T\epsilon\, p_{2+k}\pi_i - \frac{1}{2}(-1+T)\,T^3\epsilon\, p_{2+i}p_{2+j}\pi_i^2 + \frac{1}{2}(-1+T)\,T^3\epsilon\, p_{2+j}^2\pi_i^2 - \frac{1}{2}(-1+T)\,T^2\epsilon\, p_{2+i}p_{2+k}\pi_i^2 + \frac{1}{2}(-1+T)^2\,T\epsilon\, p_{2+j}p_{2+k}\pi_i^2 + \frac{1}{2}(-1+T)\,T\epsilon\, p_{2+k}^2\pi_i^2 + i\,T p_{2+j}\pi_j - i\,T\epsilon\, p_{2+j}\pi_j - i\,(-1+T)\,p_{2+k}\pi_j + i\,(-1+2T)\,\epsilon\, p_{2+k}\pi_j + T^3\epsilon\, p_{2+i}p_{2+j}\pi_i\pi_j - T^3\epsilon\, p_{2+j}^2\pi_i\pi_j - (-1+T)\,T^2\epsilon\, p_{2+i}p_{2+k}\pi_i\pi_j + (-1+T)^2\,T\epsilon\, p_{2+j}p_{2+k}\pi_i\pi_j + (-1+T)\,T\epsilon\, p_{2+k}^2\pi_i\pi_j - \frac{1}{2}(-1+T)\,T\epsilon\, p_{2+j}p_{2+k}\pi_j^2 + \frac{1}{2}(-1+T)\,T\epsilon\, p_{2+k}^2\pi_j^2 + i\,p_{2+k}\pi_k - 2\,i\,\epsilon\, p_{2+k}\pi_k + T^2\epsilon\, p_{2+i}p_{2+k}\pi_i\pi_k - (-1+T)\,T\epsilon\, p_{2+j}p_{2+k}\pi_i\pi_k - T\epsilon\, p_{2+k}^2\pi_i\pi_k + T\epsilon\, p_{2+j}p_{2+k}\pi_j\pi_k - T\epsilon\, p_{2+k}^2\pi_j\pi_k\Big]$$

Invariance under the other Reidemeister moves is proven in a similar way. See IType.nb at ωεβ/ap.

Where is it coming from? The most honest answer is "we don't know" (and that's good!). The second most, "undetermined coefficients for an ansatz that made sense". The ansatz comes from the following principles / earlier work:

Morphisms have generating functions. Indeed, there is an isomorphism

$$\mathcal{G}\colon\ \mathrm{Hom}(\mathbb{Q}[x_i],\mathbb{Q}[y_j]) \to \mathbb{Q}[y_j][\![\xi_i]\!],$$

and by PBW, many relevant spaces are polynomial rings, though only as vector spaces.

Composition is integration. Indeed, if $f \in \mathrm{Hom}(\mathbb{Q}[x_i],\mathbb{Q}[y_j])$ and $g \in \mathrm{Hom}(\mathbb{Q}[y_j],\mathbb{Q}[z_k])$, then

$$\mathcal{G}(g\circ f) = \int e^{-y\cdot\eta} f g\, dy\, d\eta$$

Use universal invariants. These take values in a universal enveloping algebra (misschien kwantized), and thus they are expressible as long compositions of generating functions. See [La, Oh].

"Solvable approximation" ⇝ perturbed Gaussians. Let $\mathfrak{g}$ be a semisimple Lie algebra, let $\mathfrak{h}$ be its Cartan subalgebra, and let $\mathfrak{b}^u$ and $\mathfrak{b}^l$ be its upper and lower Borel subalgebras. Then $\mathfrak{b}^u$ has a bracket β, and as the dual of $\mathfrak{b}^l$ it also has a cobracket δ, and in fact, $\mathfrak{g}\oplus\mathfrak{h} \equiv \mathrm{Double}(\mathfrak{b}^u,\beta,\delta)$. Let $\mathfrak{g}^+_\epsilon := \mathrm{Double}(\mathfrak{b}^u,\beta,\epsilon\delta)$ (mod ϵ^{d+1} it is solvable for any d). Then by [BV3, BN1] (in the case of $\mathfrak{g} = sl_2$) all the interesting tensors of $\mathcal{U}(\mathfrak{g}^+_\epsilon)$ (quantized or not) are perturbed Gaussian with perturbation parameter ϵ with with understood bounds on the degrees of the perturbations.

The Philosophy Corner. "Universal invariants", valued in universal enveloping algebra (possibly quantized) rather than in representations thereof, are a priori better than the representation theoretic ones. They are compatible with strand doubling (the Hopf coproduct), and as the knot genus and the ribbon property for knots are expressible in terms of strand doubling, universal invariants stand a chance to say something about these properties. Indeed, they sometimes do! See e.g. [BN2, GK, LV, BG]. Representation theoretic invariants don't do that!

There's more! To get sl_2 invariants mod ϵ^3, add the following to $L(X^+_{ij})$, $L(X^-_{ij})$, and $L(C^\varphi_i)$, respectively (and see More.nb at ωεβ/ap for the verifications):

☺ $\epsilon^2\, r_2[1, i, j]$

💻 $\frac{1}{12}\epsilon^2 \left(-6 p_i x_i + 6 p_j x_i - 3 (-1 + 3 T)\, p_i p_j x_i^2 + 3 (-1 + 3 T)\, p_j^2 x_i^2 + 4 (-1 + T)\, p_i^2 p_j x_i^3 - 2 (-1 + T) (5 + T)\, p_i p_j^2 x_i^3 + 2 (-1 + T) (3 + T)\, p_j^3 x_i^3 + 18 p_i p_j x_i x_j - 18 p_j^2 x_i x_j - 6 p_i^2 p_j x_i^2 x_j + 6 (2 + T)\, p_i p_j^2 x_i^2 x_j - 6 (1 + T)\, p_j^3 x_i^2 x_j - 6 p_i p_j^2 x_i x_j^2 + 6 p_j^3 x_i x_j^2\right)$

☺ $\epsilon^2\, r_2[-1, i, j]$

💻 $\frac{1}{12 T^2}\epsilon^2 \left(-6 T^2 p_i x_i + 6 T^2 p_j x_i + 3 (-3 + T)\, T p_i p_j x_i^2 - 3 (-3 + T)\, T p_j^2 x_i^2 - 4 (-1 + T)\, T p_i^2 p_j x_i^3 + 2 (-1 + T) (1 + 5 T)\, p_i p_j^2 x_i^3 - 2 (-1 + T) (1 + 3 T)\, p_j^3 x_i^3 + 18 T^2 p_i p_j x_i x_j - 18 T^2 p_j^2 x_i x_j - 6 T^2 p_i^2 p_j x_i^2 x_j + 6 T (1 + 2 T)\, p_i p_j^2 x_i^2 x_j - 6 T (1 + T)\, p_j^3 x_i^2 x_j - 6 T^2 p_i p_j^2 x_i x_j^2 + 6 T^2 p_j^3 x_i x_j^2\right)$

☺ $\epsilon^2\, \gamma_2[\varphi, i]$

💻 $-\frac{1}{2}\epsilon^2 \varphi^2 p_i x_i$

The sl_2 formulas mod ϵ^4 are in the last page of the handout of [BN3].

We are very close to having some sl_3 formulas, but they are certainly not ready for prime time.

References.

[BN1] D. Bar-Natan, *Everything around sl^ϵ_{2+} is DoPeGDO. So what?,* talk given in "Quantum Topology and Hyperbolic Geometry Conference", Da Nang, Vietnam, May 2019. Handout and video at ωεβ/DPG.

[BN2] D. Bar-Natan, *Algebraic Knot Theory,* talk given in Sydney, September 2019. Handout and video at ωεβ/AKT.

[BN3] D. Bar-Natan, *Cars, Interchanges, Traffic Counters, and some Pretty Darned Good Knot Invariants,* talk given in "Using Quantum Invariants to do Interesting Topology", Oaxaca, Mexico, October 2022. Handout and video at ωεβ/Cars.

[BV1] D. Bar-Natan and R. van der Veen, *A Polynomial Time Knot Polynomial,* Proc. Amer. Math. Soc. **147** (2019) 377–397, arXiv:1708.04853.

[BV2] D. Bar-Natan and R. van der Veen, *A Perturbed-Alexander Invariant,* to appear in Quantum Topology, ωεβ/APAI.

[BV3] D. Bar-Natan and R. van der Veen, *Perturbed Gaussian Generating Functions for Universal Knot Invariants,* arXiv:2109.02057.

[BG] J. Becerra Garrido, *Universal Quantum Knot Invariants,* Ph.D. thesis, University of Groningen, ωεβ/BG.

[GK] S. Garoufalidis and R. Kashaev, *Multivariable Knot Polynomials from Braided Hopf Algebras with Automorphisms,* arXiv:2311.11528.

[La] R. J. Lawrence, *Universal Link Invariants using Quantum Groups,* Proc. XVII Int. Conf. on Diff. Geom. Methods in Theor. Phys., Chester, England, August 1988. World Scientific (1989) 55–63.

[LV] D. López Neumann and R. van der Veen, *Genus Bounds from Unrolled Quantum Groups at Roots of Unity,* arXiv:2312.02070.

[Oh] T. Ohtsuki, *Quantum Invariants,* Series on Knots and Everything **29**, World Scientific 2002.

[Ov] A. Overbay, *Perturbative Expansion of the Colored Jones Polynomial,* Ph.D. thesis, University of North Carolina, August 2013, ωεβ/Ov.

[R1] L. Rozansky, *A Contribution of the Trivial Flat Connection to the Jones Polynomial and Witten's Invariant of 3D Manifolds, I,* Comm. Math. Phys. **175-2** (1996) 275–296, arXiv:hep-th/9401061.

[R2] L. Rozansky, *The Universal R-Matrix, Burau Representation and the Melvin-Morton Expansion of the Colored Jones Polynomial,* Adv. Math. **134-1** (1998) 1–31, arXiv:q-alg/9604005.

[R3] L. Rozansky, *A Universal U(1)-RCC Invariant of Links and Rationality Conjecture,* arXiv:math/0201139.

Acknowledgement. This work was supported by NSERC grant RGPIN-2018-04350 and by the Chu Family Foundation (NYC).

Disclaimer. It's fun, but not fully ready.